

## G05EWF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

G05EWF returns the next term from an autoregressive moving-average time series using a reference vector set up by G05EGF.

### 2 Specification

```
real FUNCTION G05EWF(R, NR, IFAIL)
INTEGER                NR, IFAIL
real                 R(NR)
```

### 3 Description

The routine generates the next term in the autoregressive series and stores it in a circular buffer in the reference vector. It then applies the moving-average summation and returns the result. This is equivalent to the ARMA model described under G05EGF.

### 4 References

- [1] Tunnicliffe–Wilson G (1979) Some efficient computational procedures for high order ARMA models *J. Statist. Comput. Simulation* **8** 301–309

### 5 Parameters

**1:** R(NR) — *real* array *Input/Output*  
*On entry:* the reference vector as set up by G05EGF.  
*On exit:* the updated reference vector.

**2:** NR — INTEGER *Input*  
*On entry:* the dimension of the array R as declared in the (sub)program from which G05EWF is called.

This should be the same as in the preceding call of G05EGF.

**3:** IFAIL — INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, –1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

### 6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

NR has been changed or R corrupted since it was set up by G05EGF, or since its last use by G05EWF.

## 7 Accuracy

Not applicable.

## 8 Further Comments

The time taken by the routine is of order (NA+NB), where NA is the number of autoregressive coefficients, and NB the number of moving-average coefficients, in the underlying model.

The comments made in Section 8 of the document for G05EGF, concerning the use of G05CBF, G05CCF, G05CFF and G05CGF, must be read before using this routine.

Although the reference vector may be copied like any other array, inexperienced users are strongly advised not to keep more than a single copy. Copying it at any point has the effect of starting a new, independent time series with an identical history. This facility may be useful, but it is clearly a fruitful source of confusion if misused or used by accident.

## 9 Example

This example program calls G05EGF to set up the reference vector for the autoregressive model

$$x_n = 0.4x_{n-1} + 0.2x_{n-2} + \epsilon_n$$

where  $\epsilon_n$  is a series of independent random Standard Normal perturbations. G05EWF is then called 10 times to generate a sample of observations, which are printed.

The generator mechanism used is selected by an initial call to G05ZAF.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G05EWF Example Program Text
*      NAG Fortran SMP Library, Release 2.  NAG Copyright 2000.
*      .. Parameters ..
      INTEGER          NA, NB, NR
      PARAMETER        (NA=2,NB=1,NR=NA+NB+4+NA)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*      .. Local Scalars ..
      DOUBLE PRECISION VAR, X
      INTEGER          I, IFAIL
*      .. Local Arrays ..
      DOUBLE PRECISION A(NA), B(NB), R(NR)
*      .. External Functions ..
      DOUBLE PRECISION G05EWF
      EXTERNAL          G05EWF
*      .. External Subroutines ..
      EXTERNAL          G05CBF, G05EGF, G05ZAF
*      .. Executable Statements ..
      CALL G05ZAF('0')
      WRITE (NOUT,*) 'G05EWF Example Program Results'
      WRITE (NOUT,*)
      CALL G05CBF(0)
      A(1) = 0.4D0
      A(2) = 0.2D0
      B(1) = 1.0D0
      IFAIL = 0

```

```
*
  CALL G05EGF(0.0D0,A,NA,B,NB,R,NR,VAR,IFAIL)
*
  DO 20 I = 1, 10
    IFAIL = 0
    X = G05EWF(R,NR,IFAIL)
    WRITE (NOUT,99999) X
  20 CONTINUE
  STOP
*
99999 FORMAT (1X,F12.4)
END
```

## 9.2 Program Data

None.

## 9.3 Program Results

G05EWF Example Program Results

```
2.4084
1.1987
2.4778
0.7998
0.0452
0.4125
0.3784
-1.2166
-0.3510
1.1631
```

---